
Introduction

In the last two columns, we developed a speaker engine for interfacing to simple text-to-speech software (Monologue from First Byte) and a listener engine for interfacing to speech recognition software (IBM VoiceType). Here, we develop a voice controlled recipe browser (see Figure 1) that uses both engines. The recipes were downloaded from Richard Darsie’s web site “http://itpubs.ucdavis.edu/richard/recipes.html” (see Figure 2), one of the nicer hits we found through a “recipes” web search. The recipe browser was designed to open an HTML file formatted like Richard’s file. When such an HTML file is opened, the browser strips out the HTML tags, extracts the recipe information into a collection of recipe objects, and displays the resulting recipes. A user can then select one of the recipes in the list pane with the mouse. From that point onward, the user can switch to voice control as the actual recipes are being worked on in the kitchen. With hands “wet” with cooking ingredients, we assume it’s not possible to work the mouse. By default, the browser is ready to list the ingredients via commands such as “next” or “previous”. It’s also possible to switch to the list of instructions with command “instructions” or “steps”. Commands “next” and “previous” will then apply to instructions until the user switches back to ingredients via command “ingredients” or “stuff.” 

Designing a Recipe Class

Before designing the browser, we first create a very simple Recipe class (see Listing 1) for managing an individual recipe. Each recipe has a title (a string such as 'Vegetable Curry'), an ordered collection of ingredients (the items are strings), and an ordered collection of instructions (again, strings). The get/set methods were generated automatically by VisualAge and so contain visual programming event signalling messages. Aside from a printOn: method, which we use mainly for debugging, the only other instance method is a special converter, instructionsString, which converts the collection of individual instruction strings into one multi-line string. We needed this because we decided to place the instructions in a text widget rather than a list widget.



The other major component of the Recipe class is the set of class methods for stripping away the HTML tags and recovering the detailed recipe information. Method “readAllFromHTMLFile: fileName” opens the file, extracts all the characters, and closes the file as quickly as possible. That way, we don’t have to do anything special to deal with unclosed files should subsequent errors occur. The extracted string is then handed to “extractRecipesFrom: aString” which converts it to a read stream for ease of processing. In this specially formatted HTML file, each recipe is preceded by tag '<TABLE BORDER>' (we use “skipToAll: tag” to eliminate the text preceding the initial tag and “upToAll: tag” to pick up all subsequent text). Each of these text strings makes up exactly one recipe which we further process using “recipeFromHTMLString: aString”. This method picks up the recipe title (which is between the first two of three  '<TR>' tags), along with the ingredients and instructions (which immediately follow the third tag and are themselves each preceded by  '</TD>' tags). Finally, “separateIngredients: aString” and “separateInstructions: aString” (“separate” is to be read as a verb command) respectively convert the resulting string into ingredient and instruction lists (both string collections). The former assumes that each line is one ingredient whereas the latter specifically looks for statement terminators (periods, exclamation points, and question marks). The “cleanup: aString” method eliminates all remaining tags.



Clearly, this is not a general purpose to reading recipes off the web but it is adequate for our purposes. Perhaps some of you might wish to look up more standardized recipe sources and develop corresponding readers.

The Recipe Browser

Class RecipeBrowser (see Listing 2) is a visual part with seven instance variables: listenerEngine, speakerEngine, ingredientsStream, instructionsStream, activeStream, recipes, and counter. It was designed with the composition editor (see Figure 3) to contain three widgets, a file menu, and a file prompter.  Using Figure 1 to more easily differentiate the three widgets, we can see that each recipe browser contains a drop-down list widget (for the individual recipe titles), a list widget (for the selected recipe ingredients), and a text widget (for the selected recipe instructions). The widgets are accessible via respective messages recipesWidget, ingredientsWidget, and instructionsWidget. The recipe browser also contains a “File” menu with an associated “Open …” submenu that is connected to a file prompter to permits a user to open the recipe file mentioned above. 



When an instance is created, a listener engine is created (see initialize in Listing 2) which can report listenerReady, listenerHeard, listenerDistracted, and listenerClosed for the respective listener events #opened, #heard, #distracted, and #closed (we will discuss these listener event handlers in more detail subsequently). A speaker engine is also created (this engine doesn’t need event handlers since it is so much simpler than the other engine).



In the composition editor, we made the following attribute-to-attribute connections:



The recipe widget’s “items” attribute was connected to the “recipes” attribute.



along with the following event-to-script connections:



The browser’s #openedWidget event was connected to method browserOpened which simply opens both engines.

The browser’s #closedWidget event was connected to method browserClosed which similarly closes both engines.

The browser’s #gettingFocus event was connected to method browserActivated which tells the listener engine to listen (it is intended to be in the listening state as long as the application is active but it can stop listening if the application loses focus to another application that uses IBM VoiceType).

The browser’s #losingFocus event was connected to method browserDeactivated which tells the listener engine to stop listening.

The #clicked event associated with the “Open…” submenu item in the “File” menu was connected to the prompt action of the file prompter causing an instance of the prompter to appear. The #ok event associated with the prompter was then connected to the “openFile: fileName” script/method (the file name is a parameter supplied by the event). The openFile: method uses class method readAllFromHTMLFile: in Recipe to obtain a collection of recipe objects. The recipe objects are stored in the browser (in instance variable “recipes”) and the first recipe selected (if there is one).

The #selectedItemChanged event in the drop down list widget was connected to method “updateSelectedRecipe: aRecipe” (parameter “aRecipe” comes from the event itself) which extracts the recipe ingredients and instructions and stores them into the corresponding widgets mentioned earlier. It also creates two streams  (one for ingredients and one for instructions) and initializes a special variable “activeStream” to reference one of these streams. The streams allow us to track the current ingredient (or instruction) and “activeStream” allows us to determine whether we are currently focusing on the ingredients or the instructions.



For debugging purposes, we also introduced a variable “counter” which is incremented each time the listener engine hears a word. The counter is currently displayed in the window’s title box.

Listener Event Handling

The #opened listener event handler, listenerReady, which is triggered by the #gettingFocus event, tells the listener engine what words to expect (method candidateWords constructs a list of candidates from an array of collections of synonyms provided by candidateWordLists). The corresponding #distracted listener event handler, listenerDistracted, which is triggered by the #losingFocus event, doesn’t have to do anything (the method currently contains a comment) because everything is set up once more when focus is obtained again.



Since the #closedWidget event triggers method browserClosed, which closes both engines, nothing more needs to be done once the listener engine is close. So the #closed listener event handler, listenerClosed, does nothing (it too just contains a comment).



All of the important work is done by the #heard listener event handler, listenerHeard, which updates the title in the recipe browser (for debugging purposes), and then searches for the word that was heard. Since the engine provides an empty string if it didn’t understand what was said, we test for that possibility first. Next, we search for a matching word using sequencing message candidateWordsAndSelectorsDo: which in turn provides synonyms from candidateWordLists and corresponding selectors from wordSelectors. If a match is found, the corresponding selector is executed. Otherwise, heardSomethingUnexpected is executed (but that should never happen).



Each of the corresponding heardXXX methods handles one word, “XXX”. Method heardHelp tells the user what can be said. Method heardIngredients and heardInstructions switch to the appropriate stream and simulate having heard “next”. Method heardNext highlights the next entry (more on that below), gets the speaker engine to say the contents of the next entry, and then tells the speaker engine to listen for more. A special case is made for the case where there is no next entry. Method heardRepeat backs up the stream by one entry and simulates having heard “next”. Similarly, heardPrevious backs up the stream by one entry and simulates having heard “repeat” which is equivalent to backup by two entries. Methods heardTop and heardBottom respectively reset the stream to the beginning or end and simulate having heard “next”. Finally, heardNothing is a no-op, whereas heardSomethingUnexpected queries the user. Since all responses end with either heardNext, heardNothing, or heardSomethingUnexpected, only these three methods tell the speaker engine to listen some more. Perhaps it would be better if this were done in method listenerHeard:.

Widget Highlighting

Our initial implementation provided no feedback in the widgets. After choosing a new recipe and saying “next” enough times, we would end up hearing entries that weren’t visible in the widgets. That didn’t feel right. So we extended the application to select the text that was being spoken by the speaker engine. For the ingredients widget which is a list widget, that meant selecting the appropriate entry via “selectionIndex: entryIndex”. For the instructions widget which is a text widget, that meant selecting a substring in the widget’s text. Once we have determined the start and end index of this substring, this can be done using “selectTextFrom: start to: end”. However, this wasn’t quite enough. If this entire text didn’t fit in the window, we would end up seeing the tail end of the text. By first using “cursorPosition: start”, the widget would ensure that the cursor was visible by scrolling appropriately.

Conclusions

We built a simple speech controlled application using the listener engine and the speaker engine described in the two previous issues of the Smalltalk column. In this issue, we illustrated how easy it is to design such applications when very simple engine APIs are provided. 



Unfortunately, few vendors provide simple APIs. That’s why our listener and speaker engines had to provide a simplifying abstraction layer above that of the vendors. We conjecture that vendors spend so much time worrying about the complicated situations and the exceptional circumstances that they are too exhausted to design a “poor man’s” subset that works in typical situations. Alternatively, they start with the hard cases rather than the simple cases. So they don’t have APIs for the simple cases. Perhaps the day has come when testing for bugs is not enough. We now have to test for simplicity. We need simplicity testers (software engineers not familiar with the software) who are timed to see how long it takes them to build a simple demonstration application. It took us a week to build the infrastructure (the engines) and less than a day to build an application using this infrastructure. How long should it take the simplicity testers?

Figure 1: The voice-controlled recipe browser.

Figure 2: The downloaded recipes as they appear in Netscape.

Figure 3: The recipe browser in the composition editor.



Listing 1 The Recipe class.



class:									Recipe

superclass:							Object 

instance variables:			title ingredients instructions



comment							A non-visual part



class methods



HTML processing



readAllFromHTMLFile: fileName

	"Recipe readAllFromHTMLFile: 'c:\recipes.html'"

	| fileStream everything recipes |

	fileStream := CfsReadFileStream open: fileName.

	fileStream isCfsError ifTrue: [^OrderedCollection new].

	everything := fileStream contents.

	fileStream close.

	recipes := self extractRecipesFrom: everything.

	^recipes



extractRecipesFrom: aString

	| stream recipes separator |

	stream := ReadStream on: aString.

	recipes := OrderedCollection new.

	separator := '<TABLE BORDER>'.

	stream skipToAll: separator.

	[stream atEnd] whileFalse: [

		recipes add: (stream upToAll: separator)].

	^recipes collect: [:recipe | 

		self recipeFromHTMLString: recipe] 



recipeFromHTMLString: aString

	| stream title ignoredHeadings ingredients instructions |

	stream := ReadStream on: aString.

	stream skipToAll: '<TR>'.

	title := stream upToAll: '<TR>'.

	ignoredHeadings := stream upToAll: '<TR>'.

	ingredients := stream upToAll: '</TD>'.

	instructions := stream upToAll: '</TD>'.

	^Recipe new

		title: (self cleanup: title);

		ingredients: (self separateIngredients: ingredients);

		instructions: (self separateInstructions: instructions)



separateIngredients: aString

	| ingredients |

	ingredients := (self cleanup: aString) 

		subStrings: CldtConstants::Cr.

	^ingredients collect: [:ingredient |

		ingredient trimSeparators]



separateInstructions: aString

	| stream sentence instructions |

	stream := ReadStream on: (self cleanup: aString).

	instructions := OrderedCollection new.

	[stream atEnd] whileFalse: [

		sentence := stream abrUpToAny: #($. $! $?).

		instructions add: sentence trimSeparators, 

			(stream skip: -1; next: 1)].

	^instructions



cleanup: aString

	| inputStream outputStream data |

	"Strip out <...> sequences."

	inputStream := ReadStream on: aString.

	outputStream := WriteStream on: String new.

	[inputStream atEnd] whileFalse: [

		outputStream nextPutAll: (inputStream upTo: $<); space.

		inputStream skipTo: $>].

	"Strip out &...; sequences and replace by first character."

	inputStream := ReadStream on: outputStream contents.

	outputStream := WriteStream on: String new.

	[inputStream atEnd] whileFalse: [

		outputStream nextPutAll: (inputStream upTo: $&).

		data := inputStream upTo: $;.

		(data isNil or: [data notEmpty]) ifTrue: [

			outputStream nextPut: data first]].

	^outputStream contents trimSeparators



instance methods



get/set



ingredients

	^ingredients

ingredients: anOrderedCollection

	ingredients := anOrderedCollection.

	self signalEvent: #ingredients

		 with: anOrderedCollection. 

instructions

	^instructions

instructions: anOrderedCollection

	instructions := anOrderedCollection.

	self signalEvent: #instructions

		 with: anOrderedCollection. 

title

	^title

title: aString

	title := aString.

	self signalEvent: #title with: aString. 



printing



printOn: aStream

	aStream

		nextPutAll: 'Recipe'; cr; tab;

		nextPutAll: 'Title: '; nextPutAll: title; cr; tab;

		nextPutAll: 'Ingredients:'.

	self ingredients do: [:ingredient |

		aStream cr; tab; tab; nextPutAll: ingredient].

	aStream

		cr; tab;

		nextPutAll: 'Instructions:'.

	self instructions do: [:sentence |

		aStream cr; tab; tab; nextPutAll: sentence].



querying



instructionsString

	| output |

	output := WriteStream on: (String new: 300).

	self instructions do: [:instruction |

		output nextPutAll: instruction; cr].

	^output contents





Listing 2 The RecipeBrowser class.



class:									RecipeBrowser

superclass:							AbtAppBldrView 

instance variables:			listenerEngine speakerEngine 

											ingredientsStream instructionsStream 

											activeStream recipes counter



comment							A visual part



instance methods



initializing



initialize

	listenerEngine := (ListenerEngine on: self)

		when: #opened send: #listenerReady;

		when: #heard send: #listenerHeard:;

		when: #distracted send: #listenerDistracted;

		when: #closed send: #listenerClosed.

	speakerEngine := SpeakerEngine new



get/set



recipes

	^recipes

recipes: anOrderedCollection

	recipes := anOrderedCollection.

	self signalEvent: #recipes

		 with: anOrderedCollection. 



part references



recipesWidget

	^self subpartNamed: 'Recipe List'

ingredientsWidget

	^self subpartNamed: 'Recipe Ingredients'

instructionsWidget

	^self subpartNamed: 'Recipe Instructions'



event-to-script handlers



browserOpened

	listenerEngine open.

	speakerEngine open

browserClosed

	listenerEngine close.

	speakerEngine close



browserActivated

	listenerEngine isOpen ifFalse: [^self].

	listenerEngine listen

browserDeactivated

	listenerEngine ignore.



openFile: fileName

	fileName isNil ifTrue: [^self] "cancelled"

	self recipes: (Recipe readAllFromHTMLFile: fileName).

	self recipes isEmpty ifTrue: [^self].

	self recipesWidget selectedItem: self recipes first



updateSelectedRecipe: aRecipe

	| ingredients instructions instructionsString |

	aRecipe isNil 

		ifTrue: [

			ingredients := #(). 

			instructions := #(). 

			instructionsString := '']

		ifFalse: [

			ingredients := aRecipe ingredients.

			instructions := aRecipe instructions.

			instructionsString := aRecipe instructionsString].

	self ingredientsWidget items: ingredients.

	self instructionsWidget string: instructionsString.

	ingredientsStream := ReadStream on: ingredients.

	instructionsStream := ReadStream on: instructions.

	activeStream := ingredientsStream



listener event handling



listenerReady

	listenerEngine expect: self candidateWords.

	speakerEngine say: 'Ready'.

	listenerEngine listen



listenerClosed

	"Bye yawl."



listenerDistracted

	"We'll start listening again when the application is re-activated."



listenerHeard: word

	counter isNil ifTrue: [counter := 0].

	counter := counter + 1.

	(self subpartNamed: 'Window') 

		title: 'Recipe Browser ', counter printString, 

			': heard "', word, '"'.

	word = '' ifTrue: [^self heardNothing].

	self candidateWordsAndSelectorsDo: [:candidates :selector |

		(candidates includes: word) 

			ifTrue: [^self perform: selector]].

	self heardSomethingUnexpected



listener event handling support



candidateWordLists

	"Each subarray contains synonyms."

	^#(

		('help' 'what can I say')

		('instructions' 'steps')

		('ingredients' 'stuff')

		('next' 'now' 'forward' 'ok' 'down')

		('previous' 'back' 'backup' 'up')

		('again' 'repeat' 'what')

		('top' 'first' 'restart' 'beginning')

		('bottom' 'last' 'ending'))



candidateWords 

	^self candidateWordLists 

		inject: OrderedCollection new

		into: [:list :sublist |

			list addAll: sublist; yourself]



candidateWordsAndSelectorsDo: aBlock

	self candidateWordLists 

		with: self wordSelectors 

		do: aBlock



wordSelectors

	"Each entry matches one synonym list in candidateWordLists."

	^#(

		heardHelp	

		heardInstructions

		heardIngredients

		heardNext

		heardPrevious

		heardRepeat

		heardTop

		heardBottom)



heard reactions



heardHelp	

	speakerEngine

		say: 

			'To switch to instructions, say steps or instructions.

			To switch to ingredients, say stuff or ingredients.

			To move ahead, say down, next, forward, now, or ok.

			To move back, say up, previous, backup, or back.

			To hear something again, say repeat, what, or again.

			To go back to the top, say top, first, restart, or beginning.

			To go to the bottom, say bottom, last, or ending.



			Note: you can''t change recipes orally. 

			You have to click on a new one.

			At that point, saying next will read the next ingredient.'.

	listenerEngine listen



heardIngredients

	activeStream := ingredientsStream.

	self heardNext



heardInstructions

	activeStream := instructionsStream.

	self heardNext



heardNext

	self highlight. 

	activeStream atEnd

		ifTrue: [^self handleNextWhenAtEnd].

	speakerEngine say: activeStream next.

	listenerEngine listen



handleNextWhenAtEnd

	activeStream size = 0

		ifTrue: [

			speakerEngine say: 'None'. 

			listenerEngine listen]

		ifFalse: [

			speakerEngine say: 'At the end'. 

			self heardRepeat]



heardRepeat

	activeStream position > 0 

		ifTrue: [activeStream skip: -1]

		ifFalse: [speakerEngine say: 'I haven''t said anything yet!'].

	self heardNext



heardPrevious

	activeStream position > 1 

		ifTrue: [activeStream skip: -1]

		ifFalse: [speakerEngine say: 'I''m already at the beginning'].

	self heardRepeat



heardTop

	activeStream reset.

	self heardNext



heardBottom

	activeStream position: (activeStream size - 1 max: 0).

	self heardNext



heardNothing

	listenerEngine listen



heardNothing heardSomethingUnexpected

	speakerEngine say: 'What was that?'.

	listenerEngine listen



highlighting



highlight

	self readingIngredients	

		ifTrue: [self highlightIngredients]

		ifFalse: [self highlightInstructions]



highlightIngredients

	self ingredientsWidget 

		selectionIndex: ingredientsStream position + 1



highlightInstructions

	| string start end |

	string := instructionsStream peek.

	start := (self instructionsWidget string 

		indexOfSubCollection: string 

		startingAt: 1) - 1.

	end := start + string size.

	self instructionsWidget

		cursorPosition: start;

		selectTextFrom: start to: end



readingIngredients

	^activeStream == ingredientsStream

readingInstructions

	^activeStream == instructionsStream

